



BERKELEY LAB

Bringing Science Solutions to the World



U.S. DEPARTMENT OF
ENERGY

Office of Science

Fortran 2023 for you: Features and tools

Katherine Rasmussen, Damian Rouson

NHR PerfLab Seminar

March 10, 2025

Table of Contents

01

Features of
Fortran 2023

02

Tools to aid
Fortran
development

03

Fortran developer
communities



Features

Fortran 2023/ Modern Fortran

- Subset of features in the Fortran language
- New features from Fortran standards 2003, 2008, 2018, 2023, ~2028
- Features often considered Modern Fortran include
 - Loop level parallelism: `do-concurrent`
 - Multi-image parallelism: `coarrays` and related routines (`this_image`, `ucobound`, etc)
 - Collective and atomic subroutines (`co_sum`, `atomic_fetch_add`, etc)
 - Teams, locks, and events (`form-team-stmt`, `lock-stmt`, `event-post-stmt`, etc)
 - Polymorphism
 - C-interoperability



Fortran - Natively parallel language (since F2008)

- Coarrays and more (collectives, atomics, etc)
 - SPMD (single program multiple data) parallelism
 - Uses a PGAS (partitioned global address space) shared memory abstraction
 - [PRIF](#) and [Caffeine](#) - runtime library
 - targeting LLVM Flang and LFortran
- **do-concurrent**
 - Loop-level parallelism
 - Currently 3 compilers (NVIDIA, Intel, HPE) support automatic offloading to GPUs
- Benefits include easier to write and potentially faster to run



Useful Intrinsic Functions/Features

- [findloc\(\)](#) - Find location of a specified value in an array (optional **mask** arg)
- [pack\(\)](#) - Pack an array into an array of rank one after **mask** is applied
- [count\(\)](#) - Count true values in an array
- [merge\(\)](#) - Merge variables based on the logic of the **mask** arg
- [index\(\)](#) - Find position of a substring within a string
- Note: **mask** argument (boolean logic) of intrinsic functions
- Note: Combining intrinsic function calls
- More intrinsics with explanations [here](#)



Useful Intrinsic Functions - Example - Use Case

```
An array of bin_t objects (bins)
  passes on partitioning items nearly evenly across bins.
  passes on partitioning all item across all bins without item loss.
2 of 2 tests pass.
```

```
A format string
  passes on yielding a comma-separated list of real numbers.
  passes on yielding a comma-separated list of double-precision numbers.
  passes on yielding a space-separated list of complex numbers.
  passes on yielding a comma- and space-separated list of character values.
  passes on yielding a new-line-separated list of integer numbers.
5 of 5 tests pass.
```

- Example of Julienne tests (more later)
- Feature: when running the tests, can pass a flag with a string that selects which tests are run: `-- --contains <string>`



Useful Intrinsic Functions - Example - Use Case

An array of bin t objects (bins)

```
passes on partitioning items nearly evenly across bins.  
passes on partitioning all item across all bins without item loss.  
2 of 2 tests pass.
```

A format string

subject string

array of test description strings

```
passes on yielding a comma-separated list of real numbers.  
passes on yielding a comma-separated list of double-precision numbers.  
passes on yielding a space-separated list of complex numbers.  
passes on yielding a comma- and space-separated list of character values.  
passes on yielding a new-line-separated list of integer numbers.  
5 of 5 tests pass.
```

- Need to search the subject string for a substring
- If found, run all of the tests
- If not, search the array of test description strings for the substring and run only the tests where it appears



Starting version

37 lines

```
55  block
56  integer i, m, num_descriptions, num_matches
57  logical substring_in_subject
58  logical, allocatable :: substring_in_description(:)
59  character(len=:), allocatable :: test_subject
60  num_descriptions = size(test_descriptions)
61  test_subject = subject()
62  substring_in_subject = .false.
63  do i = 1, len(test_subject) - len(test_description_substring) + 1
64      if (test_subject(i:i+len(test_description_substring) - 1) == test_description_substring) then
65          substring_in_subject = .true.
66          exit
67      end if
68  end do
69  if (substring_in_subject) then
70      allocate(test_results(num_descriptions))
71      do i = 1, num_descriptions
72          test_results = test_descriptions%run()
73      end do
74  else ! substring not found in subject
75      allocate(substring_in_description(num_descriptions))
76      num_matches = 0
77      do i = 1, num_descriptions
78          substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
79          if (substring_in_description(i)) num_matches = num_matches + 1
80      end do
81      allocate(test_results(num_matches))
82      m = 0
83      do i = 1, num_descriptions
84          if (m==num_matches) exit
85          if (substring_in_description(i)) then
86              m = m + 1
87              test_results(m) = test_descriptions(i)%run()
88          end if
89      end do
90  end if
91  end block
```

Starting version

37 lines

```
55  block
56      integer i, m, num_descriptions, num_matches
57      logical substring_in_subject
58      logical, allocatable :: substring_in_description(:)
59      character(len=:), allocatable :: test_subject
60      num_descriptions = size(test_descriptions)
61      test_subject = subject()
62      substring_in_subject = .false.
63      do i = 1, len(test_subject) - len(test_description_substring) + 1
64          if (test_subject(i:i+len(test_description_substring) - 1) == test_description_substring) then
65              substring_in_subject = .true.
66              exit
67          end if
68      end do
69      if (substring_in_subject) then
70          allocate(test_results(num_descriptions))
71          do i = 1, num_descriptions
72              test_results = test_descriptions%run()
73          end do
74      else ! substring not found in subject
75          allocate(substring_in_description(num_descriptions))
76          num_matches = 0
77          do i = 1, num_descriptions
78              substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
79              if (substring_in_description(i)) num_matches = num_matches + 1
80          end do
81          allocate(test_results(num_matches))
82          m = 0
83          do i = 1, num_descriptions
84              if (m==num_matches) exit
85              if (substring_in_description(i)) then
86                  m = m + 1
87                  test_results(m) = test_descriptions(i)%run()
88              end if
89          end do
90      end if
91  end block
```

Add `index()` - 27 lines

```
55     block
56     integer i, m, num_descriptions, num_matches
57     logical, allocatable :: substring_in_description(:)
58     num_descriptions = size(test_descriptions)
59     if (index(subject(), test_description_substring) /= 0) then
60         allocate(test_results(num_descriptions))
61         do i = 1, num_descriptions
62             test_results = test_descriptions%run()
63         end do
64     else ! substring not found in subject
65         allocate(substring_in_description(num_descriptions))
66         num_matches = 0
67         do i = 1, num_descriptions
68             substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
69             if (substring_in_description(i)) num_matches = num_matches + 1
70         end do
71         allocate(test_results(num_matches))
72         m = 0
73         do i = 1, num_descriptions
74             if (m==num_matches) exit
75             if (substring_in_description(i)) then
76                 m = m + 1
77                 test_results(m) = test_descriptions(i)%run()
78             end if
79         end do
80     end if
81 end block
```

27 lines version

```
55     block
56         integer i, m, num_descriptions, num_matches
57         logical, allocatable :: substring_in_description(:)
58         num_descriptions = size(test_descriptions)
59         if (index(subject(), test_description_substring) /= 0) then
60             allocate(test_results(num_descriptions))
61             do i = 1, num_descriptions
62                 test_results = test_descriptions%run()
63             end do
64         else ! substring not found in subject
65             allocate(substring_in_description(num_descriptions))
66             num_matches = 0
67             do i = 1, num_descriptions
68                 substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
69                 if (substring_in_description(i)) num_matches = num_matches + 1
70             end do
71             allocate(test_results(num_matches))
72             m = 0
73             do i = 1, num_descriptions
74                 if (m==num_matches) exit
75                 if (substring_in_description(i)) then
76                     m = m + 1
77                     test_results(m) = test_descriptions(i)%run()
78                 end if
79             end do
80         end if
81     end block
```

Add `count()` - 26 lines

```
55  block
56    integer i, m, num_descriptions, num_matches
57    logical, allocatable :: substring_in_description(:)
58    num_descriptions = size(test_descriptions)
59    if (index(subject(), test_description_substring) /= 0) then
60      allocate(test_results(num_descriptions))
61      do i = 1, num_descriptions
62        test_results = test_descriptions%run()
63      end do
64    else ! substring not found in subject
65      allocate(substring_in_description(num_descriptions))
66      do i = 1, num_descriptions
67        substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
68      end do
69      num_matches = count(substring_in_description)
70      allocate(test_results(num_matches))
71      m = 0
72      do i = 1, num_descriptions
73        if (m==num_matches) exit
74        if (substring_in_description(i)) then
75          m = m + 1
76          test_results(m) = test_descriptions(m)%run()
77        end if
78      end do
79    end if
80  end block
```

26 lines version

```
55  block
56    integer i, m, num_descriptions, num_matches
57    logical, allocatable :: substring_in_description(:)
58    num_descriptions = size(test_descriptions)
59    if (index(subject(), test_description_substring) /= 0) then
60      allocate(test_results(num_descriptions))
61      do i = 1, num_descriptions
62        test_results = test_descriptions%run()
63      end do
64    else ! substring not found in subject
65      allocate(substring_in_description(num_descriptions))
66      do i = 1, num_descriptions
67        substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
68      end do
69      num_matches = count(substring_in_description)
70      allocate(test_results(num_matches))
71      m = 0
72      do i = 1, num_descriptions
73        if (m==num_matches) exit
74        if (substring_in_description(i)) then
75          m = m + 1
76          test_results(m) = test_descriptions(m)%run()
77        end if
78      end do
79    end if
80  end block
```

Add `pack()` - 12 lines (when whitespace removed)

```
55  block
56    integer i
57    logical substring_in_subject
58    logical, allocatable :: substring_in_description(:)
59
60    substring_in_subject = index(subject(), test_description_substring) /= 0
61
62    allocate(substring_in_description(size(test_descriptions)))
63
64    do i = 1, size(test_descriptions)
65      substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
66    end do
67
68    test_descriptions = pack(test_descriptions, substring_in_subject .or. substring_in_description)
69  end block
70  test_results = test_descriptions%run()
```

12 lines version (when whitespace removed)

```
55  block
56    integer i
57    logical substring_in_subject
58    logical, allocatable :: substring_in_description(:)
59
60    substring_in_subject = index(subject(), test_description_substring) /= 0
61
62    allocate(substring_in_description(size(test_descriptions)))
63
64    do i = 1, size(test_descriptions)
65      substring_in_description(i) = test_descriptions(i)%contains_text(test_description_substring)
66    end do
67
68    test_descriptions = pack(test_descriptions, substring_in_subject .or. substring_in_description)
69  end block
70  test_results = test_descriptions%run()
```

**Combine and take advantage of automatic
reallocation - ? lines**

Combine and take advantage of automatic reallocation - 2 lines

```
55 test_descriptions = pack(test_descriptions, index(subject(), test_description_substring) /= 0 .or. test_descriptions%contains_text((test_description_substring)))  
56 test_results = test_descriptions%run()
```

Combine and take advantage of automatic reallocation

```
55     test_descriptions = pack( &
56         array = test_descriptions, &
57         mask = index(subject(), test_description_substring) /= 0 &
58             .or. test_descriptions%contains_text((test_description_substring))
59     test results = test_descriptions%run()
```

Useful Fortran Features - Example Review

- Fortran 95 features utilized in example:
 - `pack` intrinsic function
 - `index` intrinsic function
 - `count` intrinsic function (in intermediate step)
 - Automatic (re)allocation of arrays (F90-F2003)
 - `elemental` type-bound procedures - `run()`
 - `pure` procedure
- Code from [julienne/test/bin_test.F90](#)



Tools

fpm

- [fpm](#) (Fortran Package Manager)
 - Package manager and build system
 - Written in Fortran
 - Maintained by Fortran developer community
 - Automatically detects file dependencies
 - Supports both Fortran and C source code
 - fpm itself very easy to install
 - Available through package managers (Homebrew, etc)
 - If using gfortran13, can also compile a one file version of the **fpm** source code to install it



Unit testing

- Test driven development
- Benefits of unit tests
- [Julienne](#)
 - Lightweight unit testing framework
 - Test output can read as a specification
 - Provides diagnostic information for failures
- [Assert](#)
 - Utility that helps verify constraints
 - Useful diagnostic output from pure procedures



Julienne Example (content of video walkthrough)

- Create new fpm project with `fpm new name_of_proj`
- Add Julienne dependency to `fpm.toml` file
- Copy Julienne example from `julienne/example/example-project` into new fpm project
- Move `specimen_m.f90` into `src` dir
- Run tests with `fpm test`
- Fix purposeful error in source code so test passes



Add a test for a new function `increment()`

```
70 function check_increment() result(test_diagnosis)
71   type(test_diagnosis_t) test_diagnosis
72   type(specimen_t) specimen
73   integer, parameter :: expected_result = 8
74   integer :: actual_result
75
76   actual_result = specimen%increment(7)
77   test_diagnosis = test_diagnosis_t( &
78     test_passed = actual_result == expected_result, &
79     diagnostics_string = "expected result " // string_t(expected_result) &
80     //"", actual result " // string_t(actual_result))
81
82 end function
```

Update `results()` function to call new test

```
38 function results() result(test_results)
39   type(test_result_t), allocatable :: test_results(:)
40   type(test_description_t), allocatable :: test_descriptions(:)
41   procedure(diagnosis_function_i), pointer :: check_zero_ptr, check_increment_ptr
42   check_zero_ptr => check_zero
43   check_increment_ptr => check_increment
44
45   test_descriptions = [ &
46     test_description_t("the type-bound function zero() producing a result of 0", check_zero_ptr), &
47     test_description_t("the type-bound function increment() producing the correct incremented integer", &
48       check_increment_ptr) ]
49
50   test_descriptions = pack( &
51     array = test_descriptions, &
52     mask = test_descriptions%contains_text(test_description_substring) .or. index(subject(), &
53       test_description_substring)/=0)
54   test_results = test_descriptions%run()
55 end function
```

Julienne `main` program - call each collection of tests

```
6 program main
7   use julienne_m, only : command_line_t
8   use specimen_test_m, only : specimen_test_t
9   implicit none
10
11  type(specimen_test_t) specimen_test
12  integer :: passes=0, tests=0
13
14  call print_usage_and_stop_if_help_requested
15  call specimen_test%report(passes, tests)
16  call report_tally_and_error_stop_if_test_fails
```

Deep Learning with Fortran

- Machine learning and AI impacts on HPC
- [Fiats](#) (Functional inference and training for surrogates)
 - Alternative name: Fortran inference and training for science
 - “training and deployment of neural-network surrogate models for computational science”
 - Automatic parallelization of batch inference



Grabbag of Fortran tools

- [Codee](#) - Static analysis tool for Fortran and C/C++
 - Code correctness
 - Modernization
 - [Codee Youtube Channel](#)
 - [Codee training at NERSC](#)
- [fortran-linter](#)
- [rojff](#) - Return of JSON for Fortran - Utility to support use of JSON files in Fortran source
- [iso_varying_string](#) - An implementation of the ISO_VARYING_STRING module as proposed for the ISO standard



Grabbag of Fortran tools

- [FORD](#) - (FORtran Documentation)
 - automatic documentation generator for Fortran projects
 - Can build documentation locally, provides html files
 - Can deploy documentation in Github Actions
 - [Example Ford Documentation](#)
- Various tools to convert fixed form code to free form code
- [A grabbag of more tools](#) - Beliaevsky/Fortran-Tools



Communities & Where to Find More

Fortran Standards Committee

- International body - [WG5](#)
- Working group - [INCITS US National body](#) (informally known as J3)
- Upcoming features:
 - Generic programming (templates, etc)
 - Type-safe templates: requirements (strong concepts)
 - Must state properties of types
 - Compiler can provide error messages in template source code
 - Asynchrony: tasks, collectives
 - Standardized Fortran preprocessor



Fortran Resources & Communities

- Stereotypes about Fortran include ideas of being antiquated
- Vibrant, engaged developer community
- For more Fortran questions or to engage and **share** with the Fortran community, visit:
 - Fortran Lang: fortran-lang.org
 - [Tutorials](#), links to [playgrounds](#), [compilers info](#)
 - Helpful language information and advice
 - Link to [LFortran](#), a Fortran compiler and interpreter
 - Discourse: fortran-lang.discourse.group
 - Fortran Wiki: fortranwiki.org
- [Fortran at LBNL](#)



Thank You

Questions?

Email: fortran@lbl.gov